



Univerzitet u Sarajevu  
Elektrotehnički fakultet u Sarajevu  
Odsjek za računarstvo i informatiku



## Zadaća № Naziv zadaće

Predmet

Ime i prezime: /  
Broj indexa: /  
Grupa: /  
Datum: /

# Teorijski uvod

Algoritam koji smo odabrali za ovu temu je RBF algoritam interpolacije (Radial Basis Function Interpolation). Ovaj algoritam se zasniva na ideji da svaka poznata tačka na osnovu koje vršimo interpolaciju jednako utiče na vrijednost funkcije u svim tačkama koje su jednako udaljene od te tačke na način opisan nekom pretpostavljenom funkcijom. Formalno rečeno

$$f(\mathbf{x}) \approx \sum_{i=1}^N w_i \phi(|\mathbf{x} - \mathbf{x}_i|)$$

gdje u ovoj formuli  $f$  predstavlja funkciju koju želimo interpolirati,  $N$  označava broj poznatih tačaka na osnovu kojih vršimo interpolaciju, dok  $\mathbf{x}_i$  su upravo te tačke. Označimo sa  $y_i$  vrijednosti  $f(\mathbf{x}_i)$ . Težinski koeficijenti  $w_i$  predstavljaju koliko intenzivno na vrijednost funkcije u proizvoljnoj tački utiče udaljenost od tačke  $\mathbf{x}_i$ , dok  $\phi$  predstavlja pretpostavljenu radijalnu funkciju koja određuje na koji način udaljenost djeluje na tu istu vrijednost.

Za funkciju  $\phi$  imamo mnogo opcija mada se najčešće koristi takozvana multikvadratna funkcija koja glasi  $\phi(r) = \sqrt{r^2 + r_0^2}$ , gdje  $r_0$  predstavlja takozvani faktor skaliranja koji se tipično uzima da je veći od prosječnih udaljenosti, a manje od najvećih udaljenosti tačaka  $\mathbf{x}_i$ . Još neke funkcije koje se mogu koristiti su inverzna multikvadratna:  $\phi(r) = \frac{1}{\sqrt{r^2 + r_0^2}}$ , gausovska:  $\phi(r) = e^{-\frac{r^2}{2r_0^2}}$ , thin-plate spline:  $\phi(r) = r^2 \log(\frac{r}{r_0})$ .

Prirodno se još postavlja pitanje kako odrediti težinske koeficijente nakon što sto pretpostavili radijalnu funkciju. Uvrštavanjem vrijednosti  $\mathbf{x}_n$  u gore navedenu formulu znak  $\approx$  možemo da zamijenimo znakom jednakosti jer se radi o interpolaciji, te nam to daje

$$f(\mathbf{x}_n) = \sum_{i=1}^N w_i \phi(|\mathbf{x}_n - \mathbf{x}_i|) \quad , n = \overline{1, N}$$

a ovo je ništa drugo nego sistem od  $N$  linearnih jednačina sa  $N$  nepoznatih ( $w_i$  za  $i = \overline{1, N}$ ). Rješavanjem ovog sistema metodama za rješavanje sistema line-

arnih jednačina (na primjer Gaussova eliminacija) dobijamo tražene težinske koeficiente. Interpolacija se onda vrši po prvoj navedenoj formuli

# Implementacija algoritma

## Opis implementacije

Algoritam je pisan u Scilabu i sastoji se od sljedećih funkcija. Za funkciju  $\phi$  smo uzeli multikvadratnu funkciju.

- `radial_distance`:  
Služi za računanje udaljenosti dvije tačke n-dimenzijskom prostoru koristeći standardnu formulu  $\sqrt{\sum_{i=1}^N (a_i - b_i)^2}$ .
- `multiquadratic`:  
Računa vrijednost multikvadratne funkcije za date parametre  $r$  i  $r_0$ .
- `radial_matrix`:  
Stvara trivijalnu matricu  $R$  tako da vrijedi da je  $R_{i,j} = \|\mathbf{x}_i - \mathbf{x}_j\|$  pozivajući `radial_distance`.
- `multiquadratic_matrix`:  
Prima radijalnu matricu i primjenjuje multikvadratnu funkciju na svaki član, sa vrijednošću  $r_0$  postavljenom na geometrijsku sredinu srednjeg i najvećeg elementa radijalne matrice, time ustvari računajući vrijednosti koeficijenata uz  $w_1$  sistema koji se treba riješiti.
- `point_distance_vector`:  
Prima tačku/e u kojoj/im želimo vršiti interpolaciju, kao i tečke na osnovu kojih smo vršili interpolaciju i vrijednost koeficijenta  $r_0$  korištenog da se izračuna multikvadratna matrica i vraća horizontalni vektor, odnosno matricu u slučaju da smo vršili interpolaciju u više tačaka, koji predstavlja izračunate sve odgovarajuće vrijednosti funkcije  $\phi$ .
- `RBF_weights`:  
Prima tačke u kojima vršimo interpolaciju, i vrijednosti funkcije u tim tačkama, te rješava sistem preko kojeg određujemo vrijednosti težinskih koeficijenata koristeći gore nevedene pomoćne funkcije da formira sistem.

- **RBF\_evaluate:**  
Prima tačku/e u kojim se vrši interpolacija, tačke na osnovu kojih interpoliramo, vektor težinskih koeficijenata kao i  $r_0$  koje smo koristili da ih izračunamo, i vraća interpoliranu vrijednost, odnosno vektor ako aproksimiramo u više tačaka.
- **RBF\_one\_time\_evaluation:**  
Ova funkcija služi ako hoćemo samo jednom da interpoliramo, te samo joj prosljeđujemo tačke na osnovu kojih interpoliramo, vrijednost funkcije u njima, i tačke u kojima interpoliramo, te ona poziva **RBF\_weights** i **RBF\_evaluate** i vraća tražene vrijednosti
- **Test:**  
Prima funkciju, broj argumenata, broj tačaka na osnovu kojih vršimo interpolaciju i broj tačaka u kojima interpoliramo. Stvara matrice odredjenih veličina:  $X_{train}$  i  $X_{test}$  i popunjava ig+h sa odgovarajućim brojem nasumičnih vrijednosti između -500 i 500 i računa vrijednosti funkcije u njima, nakon toga vrši se interpolaciju u  $X_{test}$  na osnovu  $X_{train}$  i vraćaju se očekivanje(stvarne) i interpolirane vrijednosti.

## Source code

```
1 function d = radial_distance(a,b)
2     d = sqrt(sum((a-b).^2));
3 endfunction
```

```
1 function fi = multiquadratic(r,r0)
2     fi = sqrt(r.^2+r0.^2);
3 endfunction
```

```
1 function R = radial_matrix(X)
2     [nrows, ncols] = size(X);
3     R = zeros(nrows, nrows);
4     for i = 1:nrows
5         for j = (i+1):nrows
6             R(i,j) = radial_distance(X(i,:),X(j,:));
7             R(j,i) = R(i,j);
8         end
9     end
10 endfunction
```

```

1 function [M, r0] = multiquadratic_matrix(X)
2     r0 = sqrt(mean(X) * max(X));
3     M = multiquadratic(X, r0);
4 endfunction

```

```

1 function W1 = point_distance_vector(A, X, r0)
2     [xrows,xcols] = size(X);
3     [arows,acols] = size(A);
4     W1 = zeros(arows, xrows);
5     for i = 1:arows
6         for j = 1:xrows
7             W1(i,j) = radial_distance(A(i,:), X(j,:));
8         end
9     end
10    W1 = multiquadratic(W1, r0);
11 endfunction

```

```

1 function [W, r0] = RBF_weights(X,Y)
2     [W1, r0] = multiquadratic_matrix(radial_matrix(X));
3     W = linsolve(W1,-Y);
4 endfunction

```

```

1 function Est = RBF_evaluate(A, X, W, r0)
2     Est = point_distance_vector(A,X,r0) * W;
3 endfunction

```

```

1 function Est = RBF_one_time_evaluation(X, Y, A)
2     [W, r0] = RBF_weights(X,Y);
3     Est = RBF_evaluate(A, X, W, r0);
4 endfunction

```

```

1 function [y_test, y_pred] = Test(f, nargs, train_size,
2     test_size)
3     X_train = rand(train_size, nargs) * 1000 - 500;
4     X_test = rand(test_size, nargs) * 1000 - 500;
5     y_train = zeros(train_size, 1);
6     y_test = zeros(test_size, 1);
7     y_pred = zeros(test_size, 1);
8     for i=1:train_size
9         y_train(i) = f(X_train(i,:));
10    end
11    for i=1:test_size
12        y_test(i) = f(X_test(i,:));

```

```
12     end
13     y_pred = RBF_one_time_evaluation(X_train, y_train, X_test)
14     ;
15 endfunction
```

## Testiranje

Obavljene su 3 faze testiranja:

Prva faza koristi jednostavna funkciju  $f(a, b, c) = a + b + c$  sa 10 poznatih tačaka, i računanjem u 10 nepoznatih tačaka. Srednja apsolutna vrijednost greške je 27.883985, dok je standardna devijacija 27.177035. Ako isti postupak ponovimo za 100 tačaka umjesto 10 (i poznatih i traženih) Srednja vrijednost i standardna devijacija glase redom 0.7763442 i 1.0250536.

Druga faza koristi funkciju glasi  $f(a, b, c) = a * b + c$  sa 100 poznatih i 1000 traženih tačaka. Srednja vrijednost apsolutne greške i njena standardna devijacija glase 4138.8169 i 3997.0845. Iako su ove vrijednosti velike, apsolutna vrijednost funkcije u prosjeku je oko  $6 * 10^5$  te ovo nije ni 10% relativna greška.

Treća faza koristi funkciju  $f(a, b, c, d, e) = a + \sin(b) + \ln(c) + 2107$  sa 100 poznatih i 100 traženih i onda daje srednju vrijednost i standardnu devijaciju apsolutne greške da glase redom 6.8748745 i 5.1381267.